
如何使用 RS-232/RS-485 通訊串口來接收設備主動傳過來的數值

資料? 比如像是條碼機 或是 RS-232 磅秤.  
 

有些設備會用 RS-232 或 RS-485 主動傳出它所採集到的資料給 PAC，比如像 “條碼機” 或某些 

RS-232 “磅秤” 等設備，若這類設備傳出的資料格式是一個數值資料，如下。  

數字<CR>   或   數字<CR><LF>   或   數字<LF><CR> 

例如:    

3456<CR>   或   543<CR><LF>    或   -24<LF><CR> 

 
註: 符號 <CR> 指的 Enter 字元，ASCII 編碼為 13，<LF> 指的 New Line 字元, ASCII 編碼為 10。 

 

若您是使用 ISaGRAF PAC – I-7188EG、I-7188XG、I-8xx7、I-8437-80 或 I-8837-80 來連接這類的 

設備，可使用類似如下的程式碼來接收它傳過來的資料。詳細的範例請到  

http://www.icpdas.com/en/faq/index.php?kind=280#751 FAQ-066 去下載 “Demo_73.pia”。 

 

以下範例假定是用 I-8437-80 的 COM3 或 COM4 RS-232 來連接一個 RS-232 “磅秤”，此磅秤當 

有輸入訊號時會將採集到的重量 1 到 10000 克 (gram) 使用以下格式傳出來。 

 

  1<CR>    ~   10000<CR> 

 

通訊線的腳位定義: 

 

 

 

 

 

 

 

 

 

下載 FAQ-066 範例 

(採用 I-8437-80 COM4) 

磅秤 
 

TXD 
RXD 
GND 

I-8437-80  
(COM4: RS-232) 

2 RXD 
3 TXD 
5 GND 

(採用 I-8437-80 COM3) 

磅秤 
 

RXD 
TXD 
GND 

I-8437-80  
(COM3: RS-232) 

2 TXD 
3 RXD 
5 GND 

http://www.icpdas.com/en/faq/index.php?kind=280#751
http://www.icpdas.com/web/product/download/software/development_tool/isagraf/document/faq/faq066_demo_chinese.zip

ISaGRAF 專案架構: 
 

   
 
 
變數定義 :  
 

Name Type Attribute Description 
INIT Boolean Internal 初值設為 True 
Port_OK Boolean Internal Com Port 開啟成功與否 
Get_One_Data Boolean Internal 收到正確資料後，設為 True 來進行其它動作 
Step Integer Internal 內部使用 
NUM1 Integer Internal Comary_r 的回傳值 (收到幾個 byte ) 
ii Integer Internal 內部使用 
result_Gram Integer Internal 接收到的值，1 ~ 10000 克 
Tmp_val Integer Internal 內部使用 
Port_No Integer Internal 使用那個 COM port。設初值為 3 
CNT1 Integer Internal 每收到 1 個正確資料，值要加 1 
T1 Timer Internal 計時用 
Msg1 Message Internal 內部使用，長度為 255 

ST 程式 – ST1 : 

------------------------------------------------------------------------------------------------------------------------ 
(* 第 1 個 PLC scan *) 
IF  INIT  THEN 
  INIT  :=  FALSE ; 
  Port_Ok  :=  COMOPEN( Port_No , 9600 , 8 , 0 , 1 ) ;   (* 開串口為 9600, 8, N, 1 *)      
  T1  :=  T#0s ;    (* 先設為 0 *) 
  STEP  :=  0 ;     (* 先設為 0 *) 
END_IF  ; 
 
IF  Port_Ok = False  THEN   (* 若開啟串口失敗 *) 
  Return  ; 
END_IF  ; 
 
(* 處理串口動作 *) 
CASE  STEP  OF 
 
  0 :  (* Step 0 : 測試串口有無收到任何 byte 資料 *)  
     IF  COMREADY( Port_No )  THEN     
        STEP  :=  1 ;      (* 至少有一個 byte 傳進來, 設 Step 為 1 *) 
        T1  :=  T#0s ;     (* 將 T1 設為 0 *) 
        Tstart( T1 ) ;       (* 開始將 T1 計時 *) 
     END_IF  ; 
 
  1 :   (* 等待一小段時間過去, 讓磅秤把所有資料 byte 都完全傳過來*) 
     IF  T1 > T#250ms  THEN    (* 這個時間跟設備的特性有關, 有的要等久一些 *) 
       Tstop( T1 ) ;       (* 停止計時 *) 
       T1  :=  T#0s ; 
       STEP  :=  0 ;     (* 回到 Step 0 *) 
       NUM1 := COMARY_R( Port_No , 1 ) ;  (*將所有傳來的 byte 收到 1 號 byte array 內*) 
 

(* 本例假定每筆資料至少有 2 個 byte, 比如 2 克為 2<CR> , 字元 ‘2’ 與字元 <CR> 各占 
1 個 byte, 最多不超過 15 byte, 若你的設備最少 Byte 數為 3, 請改用 “NUM1 >= 3” *) 

      if  (NUM1 < 15)  and  ( NUM1 >=2 )  then   
   

(* 從 Driver 版本 I-7188EG: 2.18, I-7188XG: 2.16 , i-8xx7 : 3.20 , W-8xx7: 3.37 起才有支持 *) 
          Msg1 := Ary_Str(1 , NUM1) ;  (* 將 1 號 Byte array 內 NUM1 個 byte 轉成字串 *) 

     (* 從字串內取出 1 到數個整數值 到 1 號 Integer Array 內 *) 
          Tmp_val  :=  Msg_N(Msg1 , 1) ;    
 
          (* 若回傳 -1 表示格式錯誤, 比如: ‘A123’ , ‘5.63’ , … 等並非整數 *) 
          if  (Tmp_val > 0)  then    (* 若字串內至少含一個整數 *) 
            Tmp_val  :=  Ary_N_r(1 , 1) ;  (* 將 1 號 Integer Array 內第 1 個整數取出 *) 
 
       (* 判斷是否超出範圍 , 本例假定是 1 到 10000 公克 *) 
            If  ( Tmp_val > 0)  and  ( Tmp_val <= 10000 )  then 
              result_gram  :=  Tmp_val ;   (* 正確, 將值存放到 result_gram 內 *)                
              Get_One_Data  :=  True;     (* 設為 True, 下方才能做進一步處理 *) 
            end_if  ; 
          end_if ; 
       end_if ; 
     END_IF  ; 
END_CASE  ; 
 
if  Get_One_Data  then   (* 若為 True, 可做收到正確資料後的進一步處理 *) 
   Get_One_Data  :=  False; 
   CNT1 := CNT1+1 ;     
   (* 此處可加入要進一步處理的程式碼 *) 
   (*  …  *) 
end_if ; 
------------------------------------------------------------------------------------------------------------------------ 
 
 
如何測試? 

若沒有這個 RS-232 磅秤，可用 PC 來代替。請將 PC 的 COM1 用一條 RS-232 線接到 I-8437-80

的 COM3: RS232。接線如下: 

 

 

 

 

 

 

PC 
 

2 RXD 
3 TXD 
5 GND 

I-8437-80  
(COM3: RS-232) 

2 TXD 
3 RXD 
5 GND 

之後把 Demo_73 程式下載到 I-8437-80 控制器內，若連線下載成功，ISaGRAF “Demo_73” 會顯示

如下的表集 (Spy List) 視窗，關於 Spy List 的說明, 請參考 “ISaGRAF 使用手冊” 第 9.12 節。 
  

  
 

若有收到正確資料，“CNT1” 值會加 1。此時 “result_Gram” 即為收到的資料。 

 

PC 可開啟 Hyper Terminal 來模擬 (仿真) 成 RS232 磅秤，如下: 

 

 

 

 

(請輸入一個名稱 後 按 OK) 

 

 

http://www.icpdas.com/en/download/show.php?num=333&nation=US&kind1=&model=&kw=isagraf

(請選取 PC 要使用的 RS-232 port, 一般是 COM1 ) 

 

 
 

 

輸入本 Demo_73 範例使用的通訊串口參數，9600, 8, N, 1 , No flow control 

之後會出現如下的 Hyper Terminal 視窗。 

 

 
 

接下來請開啟 Windows “Notepad” (記事本)，輸入 123 之後按 Enter。把 123 與 Enter 鍵資料 

Copy 起來。 

 

 
 

 

然後貼到 Hyper Terminal 視窗內，如下操作。 

之後若成功，ISaGRAF Spy List 視窗的 CNT1 與 result_Gram 之值 應該會改變。 

 

 
 
 
 
 
點選連結查看更多 ISaGRAF FAQ: 
http://www.icpdas.com/en/faq/index.php?kind=280#751 
 
 
請參訪 ISaGRAF 網站，取得更多新產品資訊: 
https://www.icpdas.com/en/product/guide+Software+Development__Tools+ISaGRAF 
 
 

http://www.icpdas.com/en/faq/index.php?kind=280#751
https://www.icpdas.com/en/product/guide+Software+Development__Tools+ISaGRAF

